# An Experimental Approach to the Singular Modulus $k_{100}$

Folkmar Bornemann, March 1, 2005

*Inspired by Jonathan Borwein's* manuscript ***Review of the SIAM Hundred Digit Challenge Prepared for the Mathematical Intelligencer***.

## The minimal polynomial of $\sqrt{k_{100}}$

```
<< NumberTheory`Recognize`
k100SqrtNumerical = N[ModularLambda[√-100]^(1/4), 50];
p = Recognize[k100SqrtNumerical, 12, t]
```

$1 - 1288\,t + 20\,t^2 - 1288\,t^3 - 26\,t^4 + 1288\,t^5 + 20\,t^6 + 1288\,t^7 + t^8$

## Obtaining a Suitable Field Extension

The prime factors of 100 are 2 and 5. Therefore, a look on many of the known expressions for singular moduli suggests to try splitting the minimal polynomial $p$ over a field extension of $\mathbb{Q}[\sqrt{2}, \sqrt{5}]$. In this ground field itself, the minimal polynomial splits into quadratic factors only. A little experimentation with additional simply radicals of fourth order reveals that $\sqrt{k_{100}} \in \mathbb{Q}[\sqrt{2}, 5^{1/4}]$, though this cannot, as a real field, be the splitting field of $p$ (there are still some quadratic factors left). However, this way we obtain a rather short, though not really beautiful, radical expression for $\sqrt{k_{100}}$:

```
√k100 ==
  (k100SqrtRadical = Factor[p, Extension → {√2, 5^(1/4)}][[3]] /. t → 0)
```

$$\sqrt{k_{100}} == -161 + 114\sqrt{2} - 108\,5^{1/4} + 76\sqrt{2}\,5^{1/4} - 72\sqrt{5} - 48\,5^{3/4} + 34\sqrt{2}\,5^{3/4} + 51\sqrt{10}$$

## Beautification

The form of the radical expression for $\sqrt{k_{100}}$ suggests that, if there is a factorization into simpler expressions at all, one should look for one of the form *(well, this might be too much of hindsight for being a viable experimental approach…)*

```
k100SqrtRadicalFactored =
   a1 (a2 + √2) (a3 + √5) (a4 + √10) (a5 √2 + 5^(1/4))^2;
```

Let us give it a try and calculate the coefficients $a_1, \ldots, a_5$:

```
k100 == (k100SqrtRadicalFactored /. First@
      Solve[# == 0 & /@ Coefficient[Expand[k100SqrtRadicalFactored] -
          k100SqrtRadical /. {√2 5^(1/4) → u1, √2 5^(3/4) → u2,
          √2 → u3, √5 → u4, 5^(1/4) → u5, 5^(3/4) → u6, √10 → u7},
        Table[ui, {i, 7}]], Table[ai, {i, 5}]] // Simplify)^2
```

$$k_{100} == \left(-3+2\sqrt{2}\right)^2 \left(\sqrt{2}-5^{1/4}\right)^4 \left(2+\sqrt{5}\right)^2 \left(-3+\sqrt{10}\right)^2$$

Quite a success of the experimental approach.